AD-A058 942 CINCINNATI UNIV OHIO DEPT OF MECHANICAL ENGINEERING F/G 20/4
STUDY OF REYNOLDS STRESS EQUATION FOR PREDICTION OF FLOW CHARAC--ETC(U)
AUG 78 Y G TSUEI AFOSR-77-3311

UNCLASSIFIED AFOSR-TR-78-1268 NL

1 OF 1
ADA
058942



END
DATE
FILMED
11-78
DDC



NATIONAL BUREAU OF STANDARDS
MICROCOPY RESOLUTION TEST CHART

AD A058942

DDC FILE COPY

AFOSR-TR- 78-1268

(2) LEVEL II

FINAL REPORT

STUDY OF REYNOLDS STRESS EQUATION FOR PREDICTION OF FLOW CHARACTERISTICS OF FREE JET

Air Force Office of Scientific Research

AFOSR-77-3311

Y. G. Tsuei
Department of Mechanical Engineering
University of Cincinnati
Cincinnati, OH 45221

August 11, 1978

78 09 05 004

FINAL REPORT

STUDY OF REYNOLDS STRESS EQUATION FOR PREDICTION OF FLOW CHARACTERISTICS OF FREE JET

AIR FORCE OFFICE OF SCIENTIFIC RESEARCH

AFOSR-77-3311

Y. G. TSUEI
Department of Mechanical Engineering
University of Cincinnati
Cincinnati, OH 45221

August 11, 1978

ABSTRACT

This report concerns the prediction of the flow characteristics of an isothermal free jet. A computer program has been developed similar to that of Spalding and Patankar. Reynolds stress equations are used so that not only turbulent shearing stress, but also turbulent kinetic energy and dissipation can be calculated. This program is rather short, about 280 statements, and for a moderate number of points (usually about 15), requires only five seconds per run for the Amdahl 470/V6 computer. The results compare fairly well with experiments in two-dimensional as well as in axi-symmetric jets. It is found that the similarity assumption is only approximate. Also the results can be somewhat different for different initial input turbulence conditions. Therefore, to compare the experimental results and to interpret their accuracy, particularly when no detailed measurements are made at the jet orifice, should be done cautiously. The variations due to the assigned constants in the closure model are also briefly discussed.

TABLE OF CONTENTS

INTRODUCTION

Combustion in dump combustors is a complex process which involves mixing, mass transfer, chemical reaction as well as circulations. The investigation using the Reynolds stress equation for predicting the flow characteristics of an isothermal free jet is one of the very first steps toward understanding the mixing process. Based on the boundary layer approximation of the jet mixing and neglecting the effects of a wall, the coupled equations involving momentum, turbulent shear stress, kinetic energy and dissipation are much simplified with a proper closure model. The numerical prediction by the Spalding method with the von Mises transformation is investigated with some modification for the free jet calculation. A computer program is developed for use such that it is relatively easy for the user to read and to modify the program for his needs. Further development should include the chemical species such that free jet combustion can be predicted.

Since the AFOSR-IFP-STANFORD Conference 1968 the prediction of the turbulent flow has shifted its emphasis to turbulent field methods which involve the Reynolds shear stress turbulent energy or turbulent dissipation with varying closure models as discussed in that report [1]. In 1970, Reynolds [2] presented a brief survey of the state of the art for computation of turbulent flows. Herring and Mellor [3] developed a computer program which is used fairly widely in the United States. At Imperial College, Spalding and his coworker [4] worked over ten years on a program which is large and versatile with different turbulence models. His program is used in Europe as well as in the U.S. Because the program is large and versatile it requires some training and fluid mechanics background to use the program properly. If the user wants to modify the program for his needs, he usually encounters many difficulties. It is well known that it is difficult to read and modify a computer program, especially if it is a large and complicated one, unless the user is quite familiar with the detailed procedure. Launder [5,6] used the Reynolds stress equation in a sequence of his papers on predicting turbulent flow. His criticism is the common one i.e. there are too many constants and it is, in a sense, a complicated curve fitting technique. Nevertheless, Launder's approach does give fairly good results in general. However, there are no reports on the predictions of axi-symmetric jets which involve turbulent shear stress and kinetic energy as well as dissipation. This report is for such an investigation.

GOVERNING EQUATIONS

1. Turbulence Closure Model

For a fluid of uniform density ρ the set of differential equations governing the transport process can be written in the following form:

Equation of continuity

$$\frac{\partial U_i}{\partial x_i} = 0$$

Equations of momentum

$$\frac{DU_i}{Dt} = -\frac{1}{\rho}\frac{\partial P}{\partial x_i} + \nu \frac{\partial}{\partial x_k}\frac{\partial U_i}{\partial x_k} - \frac{\partial \overline{u_i u_k}}{\partial x_k}$$

Equations for (kinematic) Reynolds stress

$$\frac{D\overline{u_i u_j}}{Dt} = -\left[\overline{u_j u_k}\frac{\partial U_i}{\partial x_k} + \overline{u_i u_k}\frac{\partial U_j}{\partial x_k}\right] - 2\nu\,\overline{\frac{\partial u_i}{\partial x_k}\frac{\partial u_j}{\partial x_k}}$$

$$+ \overline{\frac{p}{\rho}\left(\frac{\partial u_i}{\partial x_j} + \frac{\partial u_j}{\partial x_i}\right)} - \frac{\partial}{\partial x_k}\left[\overline{u_i u_j u_k} - \nu\frac{\partial \overline{u_i u_j}}{\partial x_k} + \overline{\frac{p}{\rho}(\delta_{jk} u_i + \delta_{ik} u_j)}\right]$$

Equation for dissipation

$$\frac{D\varepsilon}{Dt} = -2\nu\frac{\partial U_i}{\partial x_k}\left(\overline{\frac{\partial u_i}{\partial x_l}\frac{\partial u_k}{\partial x_l} + \frac{\partial u_l}{\partial x_i}\frac{\partial u_l}{\partial x_k}}\right) - 2\nu\,\overline{\frac{\partial u_i}{\partial x_k}\frac{\partial u_i}{\partial x_l}\frac{\partial u_k}{\partial x_l}}$$

$$-2\,\overline{\left[\nu\frac{\partial^2 u_i}{\partial x_k \partial x_l}\right]^2} - \frac{\partial}{\partial x_k}\overline{u_k \varepsilon'} - \frac{\nu}{\rho}\frac{\partial}{\partial x_i}\overline{\left[\frac{\partial p}{\partial x_l}\frac{\partial u_i}{\partial x_l}\right]}$$

where U is the mean flow velocity of the main flow,

u_i is the fluctuation velocity, i.e. $\overline{u_i} = 0$,

$$\mathcal{E} = \nu \overline{\frac{\partial u_i}{\partial x_k} \frac{\partial u_i}{\partial x_k}}$$ is the dissipation and

$\mathcal{E}'$ is the dissipation fluctuation

Since there are more unknowns than the number of equations, some closure assumptions have to be made in order to solve the equations. Based on the information on isotropic turbulence, homogenous turbulence and pure shear flow for large Reynolds number, Hanjalic and Launder [5] propose the following form of closure for Reynolds stress and dissipation.

$$\frac{D\overline{u_i u_j}}{Dt} = -\left[\overline{u_i u_k}\frac{\partial U_j}{\partial x_k} + \overline{u_j u_k}\frac{\partial U_i}{\partial x_k}\right] - \frac{2}{3}\delta_{ij}\varepsilon$$

$$- C_1 \frac{\varepsilon}{k}\left(\overline{u_i u_j} - \delta_{ij}\frac{2k}{3}\right) + \frac{\partial U_l}{\partial x_m} A_{lj}^{mi} + \frac{\partial U_l}{\partial x_m} A_{li}^{mj}$$

$$+ C_s \frac{\partial}{\partial x_k}\frac{k}{\varepsilon}\left[\overline{u_i u_l}\frac{\partial \overline{u_j u_k}}{\partial x_l} + \overline{u_j u_l}\frac{\partial \overline{u_k u_i}}{\partial x_l} + \overline{u_k u_l}\frac{\partial \overline{u_i u_j}}{\partial x_l}\right]$$

and

$$\frac{D\varepsilon}{Dt} = - C_{\varepsilon_1}\frac{\varepsilon}{k}\overline{u_i u_k}\frac{\partial U_i}{\partial x_k} - C_{\varepsilon_2}\frac{\varepsilon^2}{k} + C_\varepsilon \frac{\partial}{\partial x_k}\left(\frac{k}{\varepsilon}\overline{u_k u_l}\frac{\partial \varepsilon}{\partial x_l}\right)$$

where A^{mi}_{lj} is a function of k, $\overline{u_i u_j}$ and a constant C .

2. Boundary Layer Approximation

A simpler version of the model for boundary layer flows results in the following set of equations in (x,y) coordinates.

$$\frac{\partial U r}{\partial x} + \frac{\partial V r}{\partial y} = 0 \qquad (1)$$

$$\frac{DU}{Dt} = -\frac{1}{\rho}\frac{dp}{dx} + \nu \frac{1}{r}\frac{\partial}{\partial y} r \frac{\partial U}{\partial y} - \frac{1}{r}\frac{\partial}{\partial y}(r\,\overline{uv}) \qquad (2)$$

$$\frac{D\overline{uv}}{Dt} = -C_1 \frac{\varepsilon}{k}\overline{uv} - C_2 k \frac{\partial U}{r\partial y} + C_s \frac{1}{r}\frac{\partial}{\partial y} \frac{k^2}{\varepsilon} r \frac{\partial \overline{uv}}{\partial y} \qquad (3)$$

$$\frac{Dk}{Dt} = -\overline{uv}\frac{\partial U}{\partial y} - \varepsilon + 0.8\, C_s \frac{1}{r}\frac{\partial}{\partial y}\left(\frac{k^2}{\varepsilon} r \frac{\partial k}{\partial y}\right) \qquad (4)$$

$$\frac{D\varepsilon}{Dt} = -C_{\varepsilon 1}\frac{\varepsilon}{k}\overline{uv}\frac{\partial U}{\partial y} - C_{\varepsilon 2}\frac{\varepsilon^2}{k} + 0.5\, C_s \frac{\partial}{r\partial y}\left(\frac{k^2}{\varepsilon} r \frac{\partial \varepsilon}{\partial y}\right) \qquad (5)$$

where $C_1 = 2.8$ and $C_2 = 0.07 C_1$

There are six constants involved in the closure model as tabulated. Thus we have five equations with five unknowns, U, V, $\overline{uv}$, k and ε .

This set of equations is parabolic and the initial conditions for U, $\overline{uv}$, k and ε are usually not available at the beginning station, hence, some guess work of their distributions must be made. It is observed that owing to the uncertainty of ε and k in the earlier stages, the range of the ratio k/ε might be quite large by simply assuming some arbitrary distributions of and k respectively. One way to overcome this is by assuming that the kinetic energy k is proportional to the dissipation ε across the stream. This idea comes from the limiting case that as ε approaches zero or k approaches zero, the ratio k/ε must approach a finite value; otherwise an artificial singularity is introduced. This assumption seems particularly appropriate for free boundary flow. From dimensional considerations, it is found that k/ε is proportional to $\delta_{0.5}/U$ for free jet flow where $\delta_{0.5}$ is the conventional half width where the velocity is half of the center velocity U_c. Thus, it appears to be proper to introduce

$$f(x) = \frac{k}{\varepsilon} = C_x\left(C + \frac{\delta_{0.5}}{U_0}\right) \tag{6}$$

which gives the asymptotic expression

$$\frac{k}{\varepsilon} \rightarrow C_x \frac{\delta_{0.5}}{U_0} \sim 5.7\, x^{3/2}$$

for the plane jet and

$$\frac{k}{\varepsilon} \sim 4.0\, x^{2}$$

for an axi-symmetric jet.

The constant C represents the adjustment of the virtual origin. If this approach is adopted, it serves the following advantages in the final form of the equations.

(i) The equation for dissipation is eliminated at the beginnning stages, thus reducing the system of five differential equations to four

(ii) Only initial profiles of U, $\overline{uv}$ and k are needed.

(iii) The sensitive constant C_{ε_1}, is avoided (at least at the beginning stages). As reported by Launder [6], C_{ε_1} in the dissipation equation is such a sensitive constant that a small change of C_{ε_1}, will result in a large change in the predictions. As a matter of fact, Launder later proposed to use the value 1.44 as compared to 1.45 as had been suggested previously.

(iv) It also provides the extension to calculate the dissipation from the original five equations by using $f(x) = C_x(C + \delta_{0.5}/U_c)$ for the development of the initial stages and then use $f(x,y) = k/\varepsilon$ directly for further downstream calculation.

NUMERICAL METHOD AND CALCULATION PROCEDURE

1. Transformation from physical coordinates (x,y) to streamline coordinates (x,ψ)

The afore-mentioned five equations for U, V, $\overline{uv}$, k and ε can be reduced further to four equations for U, $\overline{uv}$, k and ε by the von Mises transformation from physical coordinates (x,y) to streamline coordinates (x, ψ) as discussed by Pantanka & Spalding [4]. Introducing the stream function ψ and a nondimensionalized stream function $\omega \equiv \frac{\psi}{\Psi}$ where Ψ represents the streamline at the edge of the flow. Thus

$$d\psi = \rho u r\, dy \quad \text{and} \quad \left.\frac{\partial \psi}{\partial x}\right|_y = \left.\frac{\partial \omega}{\partial x}\right|_y \Psi + \omega \left.\frac{\partial \Psi}{\partial x}\right|_y$$

Finally it can be arranged in the following form in (x,ψ) coordinates.

$$\frac{\partial U}{\partial x} - \frac{1}{\Psi}\frac{d\Psi}{dx}\omega\frac{\partial U}{\partial \omega} = \frac{\partial}{\partial \omega}\frac{\rho^2 U r^2}{\sigma_3 \Psi^2}\frac{\partial U}{\partial \omega} - \frac{\rho}{\Psi}\frac{\partial r\overline{uv}}{\partial y} \qquad (7)$$

$$\frac{\partial \overline{uv}}{\partial x} - \frac{1}{\Psi}\frac{d\Psi}{dx}\omega\frac{\partial \overline{uv}}{\partial \omega} = \frac{\partial}{\partial \omega}\frac{C_s\rho^2}{\Psi^2}fUkr^2\frac{\partial \overline{uv}}{\partial \omega}$$

$$- C_1\frac{\overline{uv}}{fU} - C_2\frac{k\rho}{\Psi}r\frac{\partial U}{\partial \omega} \qquad (8)$$

$$\frac{\partial k}{\partial x} - \frac{1}{\Psi}\frac{d\Psi}{dx}\omega\frac{\partial k}{\partial \omega} = \frac{\partial}{\partial \omega}0.8\frac{C_s\rho^2}{\Psi^2}fUkr^2\frac{\partial k}{\partial \omega}$$

$$- \frac{-k}{fU} - \rho\frac{\overline{uv}}{\Psi}r\frac{\partial U}{\partial \omega} \qquad (9)$$

$$\frac{\partial \varepsilon}{\partial x} - \frac{1}{\Psi}\frac{d\Psi}{dx}\omega\frac{\partial \varepsilon}{\partial \omega} = -C_{\varepsilon_1}\frac{\varepsilon}{k}\overline{uv}r\frac{\partial U}{\partial y}$$

$$- C_{\varepsilon_2}\frac{\varepsilon}{k}\varepsilon + C_s\frac{\rho^2}{\Psi^2}r\frac{\partial}{\partial y}\left(\frac{kr}{\varepsilon/k}\frac{\partial \varepsilon}{\partial y}\right) \qquad (9a)$$

It should be noted that

(i) $$\left.\frac{\partial U}{\partial x}\right|_{\omega} \neq \left.\frac{\partial U}{\partial x}\right|_{y}$$

(ii) The stream function $\Psi(x)$ at the edge is a function of x and the expression $\frac{d\Psi}{dx}$ represents the entrainment rate, while ω remains to be of value one at the edge. The entrainment will automatically adjust the width for the growth of the boundary layer.

(iii) The four differential equations can be put in a generalized standard from as discussed in [4].

$$\frac{\partial \Phi}{\partial x} + (a + b\omega)\frac{\partial \Phi}{\partial \omega} = \frac{\partial}{\partial \omega}\left(c\frac{\partial \Phi}{\partial \omega}\right) + S \qquad (10)$$

where Φ represents U, $\overline{uv}$, k or ϵ and S represents the source term.

2. Numerical Technique

The system of equations is put in finite difference form as discussed in [4] and the final expressions are arranged as

$$A_j \cdot \Phi_{j-1} + B_j \cdot \Phi_j + C_j \cdot \Phi_{j+1} = F_j \qquad (11)$$

where the Φ_j represents the unknown quantities at downstream stations to be calculated. A, B, C and F's are constants of flow characteristics evaluated at upstream station.

It is noted that the equations are linearized so that the matrix of the coefficients is in tri-diagonal form. The solution of the equations can be put in the following form as discussed in Schlichting [7] by letting

$$\Phi_j = H_j \Phi_{j+1} + G_j \tag{12}$$

where

$$G_j = \frac{F_j - A_j G_{j-1}}{B_j + A_j H_{j-1}} \quad \text{and} \quad H_j = \frac{-C_j}{B_j + A_j H_{j-1}} \tag{13}$$

The boundary condition for free jet by using symmetry at centerline

$y = 0$: $\frac{\partial U}{\partial y} = 0$ gives $G_1 = 0$ $H_1 = 1$

$\overline{uv} = 0$ gives $G_1 = 0$ $H_1 = 0$

$\frac{\partial k}{\partial y} = 0$ gives $G_1 = 0$ $H_1 = 0$ and for $\frac{\partial \epsilon}{\partial y} = 0$ also

Thus after calculation of G_j and H_j from the center toward the boundary edge for j = 2,3.... N+1 the unknowns Φ_j are successively obtained from the edge toward the center by Equation (12).

3. Entrainment

In the marching forward calculation, the non-dimensional stream function ω is constant and at the edge $\omega = 1$. However, the stream function ψ at the edge is a function of x, and it is expected that ψ will increase as x increases due to entrainment $\frac{d\psi}{dx}$. By arbitrarily choosing a point close to the edge of the boundary and arbitrarily requiring a value of the velocity at ω_B, say U_B = fraction of the center velocity U_c, the entrainment $\frac{d\psi}{dx}$ can be evaluated from Equation (7). Alternatively, the entrainment rate is properly chosen as is the case in this program.

The computer program has been written for a free turbulent jet with output in decay of center velocity, velocity profile, turbulent shear stress, turbulent kinetic energy and dissipation. The goal was to write the program concisely so that it is easier for the user to read and modify the program for his needs. This program has been tested for sensitivity in initial conditions, constants and entrainment rate. In the program, the total momentum (FLUX) is calculated and is compared with initial total momentum (SM), the difference DM= FLUX - SM serves a check for conservation of momentum. Usually the ratio $\frac{DM}{SM}$ is about a few percent.

RESULTS AND DISCUSSIONS

Based on the report by Tsuei [8], which has been tested on laminar and turbulent two dimensional jets, this computer program is an extension to the case of the axi-symmetric jet. In order to be consistent with the previous report the notations are kept about the same. An available library plotting subroutine is utilized to present the figures in the in-line plot so that the user can obtain the figures or modify the scales without writing additional subroutines. Many figures are generated, but only a few are presented here to demonstrate the variations due to the changes of constants. In all the figures, symbols 1, 2, 3 and 4 represent the mean velocity, turbulent shearing stress, turbulent kinetic energy and dissipation respectively.* These quantities are non-dimensionalized by the center velocity with the different scales indicated. Figures 3 and 4 show the slight difference in similarity due to two stations x=21 and x=26. Figure 5 shows the center velocity decay. It is noted that the prediction is lower than measurements, particularly at the development region. However, further downstream the results compare fairly well with experimenal evidence. This discrepancy might be attributed to the effects of the initial conditions, boundary layer assumptions, assigned constants, or a small longitudinal pressure gradient. It should be emphasized that because of the deviation in the velocity decay, the other non-dimensional quantities such as turbulent shear, kinetic energy, and dissipation are all

* BECAUSE OF IN-LINE PLOTTING, THE SYMBOLS 4,3,2 AND 1 MAY FALL ON TOP OF EACH OTHER

affected. It also should be noted that y/x is deliberately chosen as the coordinate since the $y/y_{a.s}$ representation fixes one point of y which make the comparison look better than it might otherwise have been. Figures 6 and 7 show the difference due to one constant C_s. Figures 8 and 9 show the spread of the width due to the entrainment rate. From these figures, it is noted the results can be different for different constants. A comparison with experiments reported by Craig [10] is shown in Figure 10. In conclusion, the results vary with the values of different combination of constants and initial conditions; however, it can compare fairly well with experiments provided the constants are chosen properly.

FURTHER DEVELOPMENT AND SUGGESTIONS

Since a computer program has been satisfactorily developed for predicting the flow characteristics such as velocity, turbulent shear, kinetic energy, and dissipation in a free jet, it is recommended that further investigation with more elaborate closure models to study the three components of turbulent intensities as well be carried out. Initial investigation should focus on the two dimensional case because most of the constants are determined by two dimensional considerations and measurements. After that, the study should be extended to the axi-symmetric jet.

ACKNOWLEDGEMENTS

The author acknowledges the support from AFOSR Col. David E. England, Col. Lowell W. Ormand and Dr. Brian Quinn. The author would also like to acknowledge the opportunity to work at the Ramjet Technology Branch, Air Force Aero-Propulsion Laboratory, Wright-Patterson Air Force Base and to express his appreciation for the discussions with Dr. R.R. Craig and Dr. F.D. Stull.

REFERENCES

1. Kline, S.J., Morkovin, M.V. Sovran, G., Cockrell, D.J. "Computation of Turbulent Boundary Layers" AFOSR-IFP - STANFORD CONFERENCE 1968, Department of Mechanical Engineering, Stanford University, Stanford, California 94305.

2. Reynolds, W.C., "Computation of Turbulent Flows - State-of-the-Art, 1970" Report MD-27, Department of Mechanical Engineering, Stanford University, Stanford, California 94305.

3. Herring, H.J. and Mellor, G.L. "A Method of Calculating Compressible Turbulent Boundary Layers" NASA CR-1444, 1968.

4. Patankar, S. & Spalding, D.E., "Heat & Mass Transfer in B.L.", Intertext Books, 1970, 2nd Edition.

5. Hanjalic, K. and Launder, B.E., "A Reynolds Stress Model of Turbulence and its Application to Thin Shear Flows", J. Fluid Mech., Vol. 52, Part 4, pp 609-638, 1972.

6. Launder, B.E., "Progress in Modeling of Turbulent Transport", College of Engineering, Penn State University, Supplementer Note July 28 - August 1, 1975.

7. Schlichting, H., "Boundary Layer Theory", 6th Edn. McGraw-Hill, N.Y.

8. Tsuei, Y.G., "Study of Reynolds Stress Equation for Prediction of Flow Characteristics of Free Jet" Final Report USAF-ASEE Summer Faculty Research (WPAFB) September 1976, Department of Mechanical Engineering, University of Cincinnati, 45221.

9. Harsha, P.T., "Free Turbulent Mixing: A Critical Evaluation of Theory and Experiment". AEDC-TR-71-36, ARO Inc., 1971.

10. Craig, R.R., "Turbulence in Free Diffusion Flames of Hydrogen-Nitrogen Mixtures Burning in Still Air" AFAPL-TR-74-68.

TABLE

CONSTANT	VALUE	BASIS FOR CHOICE	REMARKS
C_1	2.8	Return to isotropy of distorted turbulence Rotta (1962)	Launder et al (1975) $C_1 = 1.5$
C_2	0.45	Plane homogeneous shear flow Champagne et al (1970)	The value of 0.4 was later proposed
C_S	0.08		The value of 0.11 was later suggested
C_ϵ	1.45	Near-wall turbulence	Very sensitive constant, small change results in large deviation Launder later proposed to use the value 1.44
C_{ϵ_1}	2.0	Decay of grid turbulence	The value of 1.90 was proposed later by Launder et al.
C_{ϵ_2}	0.13		Launder et al (1975) recommend the value of 0.15 to be consistant with a value of κ of about 4.]


Constants:
C_1 ~ C1 = 2.8
C_2 ~ C2 = 0.07*C1
C_S ~ CS = 0.08
π ~ PI = 3.14159
C_ϵ ~ CEPS = 1.45
C_{ϵ_1} ~ CEPS1 = 2.0
C_{ϵ_2} ~ CEPS2 = 0.13
Initial Conditions:
Read: Y1, U1, T1, K1
Boundary Conditions:
GU(1) = GT(1) = GK(1) = GE(1) = HT(1) = 0
HU(1) = HK(1) = HE(1) = 1
Calculate:
Initial flow rate Q
Initial momentum flux SM
Non-dimensional stream function W
DO 9990 KN = 1, KKK
This is the main loop X = X + DX
DO 444 ITER = 1, NITER
Iterations, NITER = 6 for KN ≤ 3 and then NITER = 3
DO 120 J = 2, NP
Calculation of G's and H's
120
DO 330 JJ = 1, N
Calculation of U, T, K, E
Divergent?
Stop
330
444
Calculate entrainment ENTRN and Ψ increment DPSI
DO 440 J = 2, NPP
Calculate Y, Y5, Q and momentum flux FLUX
440
Check conservation of momentum DM = FLUX - SM
Change DX as X increases
CALL OUTPUT
9990
9999
More
Run or Data?
yes
Go to the starting point
No
11111
STOP

```
      REAL K, KI,KE
      EXTERNAL FUNC
      COMMON W(25),DW(25),Y(25),U(25),E(25),T(25),K(25),TDU(25),UDK(25)
      COMMON KN,NM,KKK,NC,NP,N,ENTRN,X,Q,PSI,SM,FLUX,DM,Y5,R,JR,F
      REAL UPE(25), GE(25), HE(25),    DY(25), YU(25), YUU(25), YY(25),
     1  KK(25), DU(25), UU(25), GU(25), HU(25), UPU(25),
     2           DT(25), TT(25), GT(25), HT(25), UPT(25), UK(25), EE(25),
     3   V(25), DK(25), DE(25), GK(25), HK(25), UPK(25),VK(25),
     4  XX(200)/200*0.0/, UC(200)/200*0.0/, GRAPH(400)/400*0.0/
      DIMENSION  YI(25), UI(25),TI(25),VI(25),EI(25),KI(25)
      REAL YS(25),US(25),TS(25),KS(25),ES(25)
      REAL XXA(8)/4., 5., 6., 8., 10., 15., 20., 25./
      REAL UCA(8)/1.0, 0.96, 0.84, 0.70, 0.57, 0.39, 0.29, 0.23/
      DATA C/0.0/, CX /5.72/,EDGE/10.0/,XGROW/0.0025/,  DELTA/0.10/
      DATA C1/2.8/,C2/0.45/,CS/0.08/,CEPS1/1.45/,CEPS2/2.0/,CEPS/0.13/
      DATA CC/0.07/, C3/0.00/, C4/0.00/, CM/0.00/, CONST/0.8/,NWRITE/1/
      DATA PI/3.14159/, PII/6.28318/,KNN/10/,RHO/1./,RJET/1./,UJ/1./
      DATA LL/0/
      DATA NPP/15/,YI/0.00,0.1, 0.2, 0.3, 0.4, 0.5, 0.6, 0.7,  .8,
     1 0.85, 0.9,  0.95, 1.0, 1.05,1.1, 10*0.0/
      DATA UI/9*1.0, 0.98, 0.96,0.90, 0.5, 0.25, 0.0, 10*0.0/,JJJJ/2/
      DATA TI/0.0, 13*0.001, 0.0, 10*0.0/, KI/25*0.001/, EI/25*0.002/
      DATA TT(1),GU(1),GT(1),GK(1),HT(1),YU(1),YUU(1),PRESS/8*0.0/
      DATA V(1),UK(1),YY(1),GE(1),  A   ,DPSI/6*0.0/,XLAST/60./,DX/.05/
      DATA HE(1),HU(1),HK(1),UU(1)/4*1.0/,UEDGE/0./,NNN/10/,LAMIN/1/
      CEPS1=1.44
      C1=2.6
      N=NPP-2
      NP=N+1
      NM=N-1
      NC=NP/2
      C1=C1+0.1
      CS=0.08
      DO 11111 LLLL=1,2
      CEPS2=CEPS2-0.1
      CS=0.08
 6666 FORMAT (/)
      DO 11111 KKKKK=1,2
      CS=CS+0.01
      DO 11111 JJJJJ=1,JJJJ
  100 CONTINUE
      ENTRN=0.5
      XGROW=-0.0002
      NNN=10
      NWRITE=1
      JR=1
      R=400.
      RR=R/1000.
      F=CX*(C+1.0)
      Y5=1.
```

```
      KI(NPP)=0.0
      DO 10 J=1,NPP
      Y(J)=YI(J)
      U(J)=UI(J)
      T(J)=TI(J)
      K(J)=KI(J)
      E(J)=EI(J)
   10 CONTINUE
      HU(1)=1.0001
      W(1)=0.0
      TDU(1)=0.0
      X=0.0
      DX=0.05
      DM=0.0
      FLUX=0.0
      KKK=68
      WRITE (6,6666)
      Q=0.0
      SM=0.0
      DO 102 J=2,NPP
      JM=J-1
      DY(JM)=Y(J)-Y(JM)
      YRM=0.5*(Y(J)+Y(JM))*2.*PI
      IF (JR.EQ.0) YRM=1.
      DW(JM)=(U(J)+U(JM))/2.*DY(JM)*YRM
      W(J)=W(JM)+DW(JM)
      Q= Q + 0.5*(U(J)+U(JM)) *DY(JM)*YRM
      SM=SM+ 0.5*(U(J)**2+ U(JM)**2)*DY(JM)*YRM
  102 CONTINUE
      SMM=SM
      DO 110 J=1,NP
      W(J)=W(J)/W(NPP)
      DW(J)=DW(J)/W(NPP)
  110 CONTINUE
      W(NPP)=1.0
      PSI=Q
  101 CONTINUE
      DO 9990 KN=1,KKK
      ENTRN=ENTRN+XGROW*X
      NPPS=NPP
      IF (KN.LE.3) NITER=6
      IF (KN.GT.3) NITER=3
      XS=X
      X=X+DX
      DO 103 J=1,NPP
      VK(J)=K(J)
  103 V(J)=U(J)
      LAMIN=0
      IF (LAMIN.EQ.1) R=30
      DO 444 ITER=1,NITER
```

```
DO 120 J=2,NP
JM=J-1
JP=J+1
UEDGE=0.0
UCE=U(1)-UEDGE
F=CX*(C+Y5*UCE)
IF ((KN.GT.KNN).AND.(LAMIN.EQ.0))  F=K(J)/E(J)
YP=Y(JP)*PII
YJ=Y(J)*PII
YM=Y(JM)*PII
DWPM=W(JP)-W(JM)
A=0
B=-ENTRN/PSI
RP=RHO/PSI
RPP=4.0*RP*RP
CMU=RPP/R*(U(JM)*YM*YM+U(J)*YJ*YJ)
CPU=RPP/R*(U(JP)*YP*YP+U(J)*YJ*YJ)
CMT=CS*RPP*F*(U(JM)*T(JM)*YM*YM+U(J)*T(J)*YJ*YJ)
CPT=CS*RPP*F*(U(JP)*T(JP)*YP*YP+U(J)*T(J)*YJ*YJ)
CMT=CS*RPP*F*(U(JM)*K(JM)*YM*YM+U(J)*K(J)*YJ*YJ)
CPT=CS*RPP*F*(U(JP)*K(JP)*YP*YP+U(J)*K(J)*YJ*YJ)
CMK=CONST*CMT
CPK=CONST*CPT
CME=0.5*CEPS*CMT/CS
CPE=0.5*CEPS*CPT/CS
G1=(DW(J)/DX+4.*A+B*(W(JP)+3.*W(J)))/4./DWPM
G2=(3./DX-B)/4.
G3= (DW(JM)/DX-4.*A-B*(W(JM)+3.*W(J)))/4./DWPM
G5U=CPU/DWPM/DW(J)
G5T=CPT/DWPM/DW(J)
G5K=CPK/DWPM/DW(J)
G5E=CPE/DWPM/DW(J)
G6U=CMU/DWPM/DW(JM)
G6T=CMT/DWPM/DW(JM)
G6K=CMK/DWPM/DW(JM)
G6E=CME/DWPM/DW(JM)
AU=G3-G6U
AT=G3-G6T
AK=G3-G6K
AE=G3-G6E
BU=G2+G5U+G6U
BT=G2+G5T+G6T+C1/F/U(J)
BK=G2+G5K+G6K+1./F/U(J)
BE=G2+G5E+G6E+CEPS2/F/U(J)
CU=G1-G5U
CT=G1-G5T
CK=G1-G5K
CE=G1-G5E
IF (ITER.GT.1) GO TO 111
UPU(J)=(DW(JM)*U(JM)+3.*DWPM*U(J)+DW(J)*U(JP))/4./DX/DWPM
```

```
      UPT(J)=(DW(JM)*T(JM)+3.*DWPM*T(J)+DW(J)*T(JP))/4./DX/DWPM
      UPK(J)=(DW(JM)*K(JM)+3.*DWPM*K(J)+DW(J)*K(JP))/4./DX/DWPM
      UPE(J)= (DW(JM)*E(JM)+3.*DWPM*E(J)+DW(J)*E(JP))/4./DX/DWPM
  111 CONTINUE
      DUDW=(U(JP)-U(JM))/DWPM
      DTDW=(T(JP)*YP-T(JM)*YM)/DWPM
      TJJM=(T(J)-T(JM))/DW(JM)*(YJ+YM)/2.
      TJPJ=(T(JP)-T(J))/DW(J)*(YP+YJ)/2.
      DTDW=(TJPJ+TJJM)/2.
      SU=PRESS-RP*DTDW
      ST=     -C2*RP*K(J)*DUDW *YJ
      SK=      -RP*T(J)*DUDW*YJ
      SE=-CEPS1*RP*T(J)/F*DUDW*YJ
      FU=UPU(J)+SU
      FT=UPT(J)+ST
      FK=UPK(J)+SK
      FE=UPE(J)+SE
      GU(J)= (FU-AU*GU(JM))/(BU+AU*HU(JM))
      HU(J)=-CU/(BU+AU*HU(JM))
      GT(J)=(FT-AT*GT(JM))/(BT+AT*HT(JM))
      HT(J)=-CT   /(BT+AT*HT(JM))
      GK(J)=(FK-AK*GK(JM))/(BK+AK*HK(JM))
      HK(J)=-CK   /(BK+AK*HK(JM))
      GE(J)=(FE-AE*GE(JM))/(BE+AE*HE(JM))
      HE(J)=-CE/(BE+AE*HE(JM))
  120 CONTINUE
      DO 330 JJ=1,NP
  130 U(NPP-JJ)=HU(NPP-JJ)*U(NPP-JJ+1)+GU(NPP-JJ)
  230 T(NPP-JJ)=HT(NPP-JJ)*T(NPP-JJ+1)+GT(NPP-JJ)
      K(NPP-JJ)=HK(NPP-JJ)*K(NPP-JJ+1)+GK(NPP-JJ)
      E(NPP-JJ)=HE(NPP-JJ)*E(NPP-JJ+1)+GE(NPP-JJ)
  330 CONTINUE
      U(NPP)=U(NP)/EDGE
      T(NPP)=T(NP)/EDGE
      K(NPP)=K(NP)/EDGE
      E(NPP)=E(NP)/EDGE
      Q=0.0
      FLUX=0.0
      DO 440 J=2,NPP
      JM=J-1
      YRM=0.5*(Y(J)+Y(JM))*2.*PI
      IF (JR.EQ.0) YRM=1.
      DY(JM)=2.*PSI*DW(JM)/(U(J)+U(JM))/YRM
      Y(J)=Y(JM)+DY(JM)
      Q= Q + 0.5*(U(J)+U(JM)) *DY(JM)*YRM
      FLUX=FLUX + 0.5*(U(J)**2+ U(JM)**2)*DY(JM)*YRM
      UU(J)=(U(J)-UEDGE)/UCE
      IF (.NOT.(UU(J).LT.0.5.AND.UU(JM).GT.0.5))  GO TO 435
      Y5=Y(JM)+(0.5-UU(JM))/(UU(J)-UU(JM))*(Y(J)-Y(JM))
  435 CONTINUE
```

```
      IF (J.EQ.NPP) GO TO 440
      JP=J+1
      DWPM=W(JP)-W(JM)
      ENTU=ENTRN *W(J)/PSI*(U(JP)-U(JM))/DWPM
      ENTK=ENTRN *W(J)/PSI*(K(JP)-K(JM))/DWPM
      TDU(J)=U(J)*((U(J)-V(J))/DX-ENTU)
      UDK(J)=U(J)*((K(J)-VK(J))/DX-ENTK)*Y5/UCE**3
  440 CONTINUE
      DM=FLUX-SM
      IF ((U(NP).GT.U(N)).AND.(T(NP).GT.T(N))) GO TO 9999
  444 CONTINUE
      UC(KN)=U(1)
      XX(KN)=X/2.
      IF (XX(KN).LE.1.1) XX(KN)=1.1
      DPSI=ENTRN*DX
      PSI=PSI+DPSI
      IF (KN.EQ.1) WRITE (6,1100)
 1100 FORMAT (1X,' KN     X     PSI   YNPP   Y5   ENTRN   DM   U(1)  U(3)  U
     1(4)  U(5)   U(6)  UNM   UN   UNP  T(NC)  T(N)  K(1)   KC    KNM
     2KN   KNP')
      IF ((KN.LE.1).OR.(KN.EQ.NWRITE)) CALL OUTPUT (5)
      IF (KN.GT.51) NNN=5
      IF (KN.EQ.NWRITE)  NWRITE=NWRITE+NNN
  480 CONTINUE
  490 CONTINUE
      IF (X.GE.1.0) DX=DELTA*X
      IF (DX.GT.2.0)  DX=2.0
      IF (JJJJJ.EQ.2.AND.X.LE.5.0) GO TO 556
      DO 555 J=1,NPP
      U(J)=U(J)*SQRT(SM/FLUX)
  555 CONTINUE
  556 CONTINUE
      SMM=FLUX
      IF (.NOT.((KN.EQ. 63).OR.(KN.EQ. 82).OR.(KN.EQ.KKK))) GO TO 9990
      DO 1 J=1,NP
      GRAPH(J)=Y(J)/X
      IF (GRAPH(J).GT.0.16) GRAPH(J)=0.16
      GRAPH(J+NP)=U(J)/U(1)
      GRAPH(J+2*NP)=T(J)/U(1)**2*40.
      GRAPH(J+3*NP)=K(J)/U(1)**2*10.
      GRAPH(J+4*NP)=E(J)*Y5/U(1)**3*40.
    1 CONTINUE
      CALL PLOT9 (1,GRAPH,NP,5,0,FUNC, .2 , 0.0,1.4, 0., 400)
      CALL OUTPUT (6)
      WRITE (6,7770)
      WRITE (6,7771) JR, N, KKK,EDGE, CEPS1, Q ,SM,FLUX, DM,  X  , F, C,
     1  CX, C1, C2, CS, CONST, DX ,RR,ENTRN
      WRITE (6,7772) CEPS,CEPS2
      WRITE (6,6666)
      CALL OUTPUT (5)
```

```
 9990 CONTINUE
      KNM=KN-1
      KNM8=KNM+8
      DO 2 J=1,KNM
      GRAPH(J+8)=XX(J)
    2 GRAPH(J+KNM8+8)=UC(J)
      DO 3 J=1,8
      GRAPH(J)=XXA(J)
    3 GRAPH(J+KNM8)=UCA(J)
      CALL PLOT9 (1,GRAPH,KNM8,2,0,FUNC, 28., 0.0,1.4, 0., 400)
      KNP=KN+1
 9999 CONTINUE
  700 WRITE (6,7770)
 7770 FORMAT (/,'   JR   N KKK  EDGE    CEPS1   Q       SM     FLUX    DM
     1     X     F       C        CX       C1       C2       CS      CONST   DX    R/
     21000 ENTRN ')
      RR=R/1000.
      WRITE (6,7771) JR, N, KKK,EDGE, CEPS1, Q ,SM,FLUX, DM,  X  , F, C,
     1  CX, C1, C2, CS, CONST, DX ,RR,ENTRN
 7771 FORMAT (1X,3I4,17F7.2)
      WRITE (6,6666)
      WRITE (6,7772) CEPS,CEPS2
 7772 FORMAT (1X,'CEPS=',F5.3,'    CEPS2=',F5.3)
11111 CONTINUE
      STOP
      END
```

```
      SUBROUTINE OUTPUT (ICALL)
      REAL K, KK, KI
      COMMON W(25),DW(25),Y(25),U(25),E(25),T(25),K(25),TDU(25),UDK(25)
      COMMON KN,NM,KKK,NC,NP,N,ENTRN,X,Q,PSI,SM,FLUX,DM,Y5,R,JR,F
      DIMENSION  YY(25),TT(25),KK(25),EE(25),UU(25)
      NPP=N+2
 1111 FORMAT (I4,1P11E10.2, 1P2E9.1)
      UEDGE=0.0
      UCE=U(1)-UEDGE
      UCEN=UCE/(1.0-UEDGE)
      Z=X/2.
      GO TO (100,200,200,200,500,600), ICALL
  100 WRITE (6,1000)
 1000 FORMAT (1X,'     INITIAL AND BOUNDARY CONDITIONS FOR  NPP  W  DW  Y
     1  U  T  K         ')
 1112 FORMAT (1X,I4,25F5.2)
      WRITE (6,1112) NPP, (W(J),J=1,NPP)
      WRITE (6,1112) N  , (DW(J),J=1,NP)
      WRITE (6,1112) NPP, (Y(J),J=1,NPP)
      WRITE (6,1112) NPP, (U(J),J=1,NPP)
      WRITE (6,1112) NPP, (T(J),J=1,NPP)
      WRITE (6,1112) NPP, (K(J),J=1,NPP)
      WRITE (6,6666)
 6666 FORMAT (/)
  200 RETURN
  500 WRITE (6,1234) KN,Z,PSI, Y(NPP), Y5, ENTRN, DM, UCEN,U(3), U(4),
     1   U(5), U(6), U(NM), U(N), U(NP),T(NC),T(N),  K(1), K(NC),
     2  K(NM), K(N), K(NP)
 1234 FORMAT  (I4,14F6.2,7F6.3)
      RETURN
  600 WRITE (6,6660)
 6660 FORMAT (/,1X,' J     W          Y          U          T          K
     1      RX         YY         UU         TT         KK         EE        TDU
     2      UDK ')
      DO 610 J=2,NP,2
      RX=Y(J)/X
      UU(J)=U(J)/UCE
      YY(J)=Y(J)/Y5
      TT(J)=T(J)/UCE**2
      KK(J)=K(J)/UCE**2
      EE(J)=E(J)*Y5/UCE**3
  610 WRITE (6,1111) J,W(J), Y(J), U(J), T(J),K(J), RX  , YY(J), UU(J),
     1  TT(J), KK(J), EE(J), TDU(J), UDK(J)
      WRITE (6,6666)
      RETURN
      END


      SUBROUTINE FUNC (XIN,YOUT)
      YOUT=0.
      RETURN
      END
```



FIG.1 NOTATION AND COORDINATES OF FREE JET



FIG.2 COORDINATES OF TRANSFORMATION



FIG.3 PROFILES OF VELOCITY, SHEAR, ENERGY AND DISSIPATION



FIG.4 PROFILES OF VELOCITY, SHEAR, ENERGY AND DISSIPATION



FIG.5 DECAY OF CENTER VELOCITY



FIG.6 PROFILES OF VELOCITY, SHEAR, ENERGY AND DISSIPATION



FIG.7 PROFILES OF VELOCITY, SHEAR, ENERGY AND DISSIPATION



FIG.8 PROFILES OF VELOCITY, SHEAR, ENERGY AND DISSIPATION



FIG.9 PROFILES OF VELOCITY, SHEAR, ENERGY AND DISSIPATION



FIG.10 COMPARISON WITH EXPERIMENTS

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER AFOSR-TR- 78 - 1268	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) STUDY OF REYNOLDS STRESS EQUATION FOR PREDICTION OF FLOW CHARACTERISTICS OF FREE JET		5. TYPE OF REPORT & PERIOD COVERED FINAL 1 Jul 77 - 30 Jun 78
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Y. G. TSUEI		8. CONTRACT OR GRANT NUMBER(s) AFOSR-77-3311
9. PERFORMING ORGANIZATION NAME AND ADDRESS UNIVERSITY OF CINCINNATI MECHANICAL & INDUSTRIAL ENGINEERING CINCINNATI, OH 45221		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 2307/09 61102F
11. CONTROLLING OFFICE NAME AND ADDRESS AIR FORCE OFFICE OF SCIENTIFIC RESEARCH/NA BLDG 410 BOLLING AIR FORCE BASE, D C 20332		12. REPORT DATE Aug 78
		13. NUMBER OF PAGES 39
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)		15. SECURITY CLASS. (of this report) UNCLASSIFIED
		15a. DECLASSIFICATION DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

TURBULENT JET
REYNOLDS STRESSES
DISSIPATION

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This report concerns the prediction of the flow characteristics of an isothermal free jet. A computer program has been developed, similar to that of Spalding and Patankar. Reynolds stress equations are used so that not only turbulent shearing stress, but also turbulent kinetic energy and dissipation can be calculated. This program is rather short, about 280 statements, and for a moderate number of points (usually about 15), requires only five seconds per run for the Amdahl 470/V6 computer. The results compare fairly well with experiments in two-dimensional as well as in axi-symmetric jets. It is found that the similarity

assumption is only approximate. Also the results can be somewhat different for different initial input turbulence conditions. Therefore, comparison of experimental results and interpretation of their accuracy, particularly when no detailed measurements are made at the jet orifice, should be done cautiously. The variations due to the assigned constants in the closure model are also briefly discussed.